Link to Mathematica .nb file

Definitions

Wigner D matrix

```
Dmn[l_, m_, n_, α_, β_, γ_] := If[β == 0,
   Limit[ⅇ^(-ⅈ (α n + γ m)) (-1)^(n+m) Sqrt[(l - m)! (l + m)! (l - n)! (l + n)!]
     Sum[(-1)^k (Cos[b/2]^(2 l+m-n-2 k) Sin[b/2]^(-m+n+2 k))/(k! (l + m - k)! (l - n - k)! (n - m + k)!), {k, 0, l+m}], b → 0]
  , ⅇ^(-ⅈ (α n + γ m)) (-1)^(n+m) Sqrt[(l - m)! (l + m)! (l - n)! (l + n)!]
    Sum[(-1)^k (Cos[β/2]^(2 l+m-n-2 k) Sin[β/2]^(-m+n+2 k))/(k! (l + m - k)! (l - n - k)! (n - m + k)!), {k, 0, l+m}]];
BigD[l_, α_, β_, γ_] := Table[Dmn[l, m, n, α, β, γ], {m, -l, l}, {n, -l, l}];
```

Complex to real transform matrix

```
Umn[l_, m_, n_] :=
  1/Sqrt[2] (KroneckerDelta[m, n] + KroneckerDelta[m, -n]) *
   Which[
    n < 0, ⅈ,
    n == 0, Sqrt[2]/2,
    n > 0, 1
   ] *
   Which[
    (m > 0) && (n < 0) && (EvenQ[n]), -1,
    (m > 0) && (n > 0) && (OddQ[n]), -1,
    True, 1];
BigU[l_] := Table[Umn[l, m, n], {m, -l, l}, {n, -l, l}];
```

Rotation matrix for order l

```
Rl[l_, α_, β_, γ_] := Re[LinearSolve[BigU[l], Dot[BigD[l, α, β, γ], BigU[l]]]]
```

Rotation matrix for all orders up to l_{max}

```
R[lmax_, α_, β_, γ_] :=
  Fold[ArrayFlatten[{{#, 0}, {0, #2}}] &, {{1}}, Table[Rl[l, α, β, γ], {l, 1, lmax}]];
```

Simple examples

Rotate $Y_{1,-1}$ 90 degrees clockwise about the z axis

```
y = {0, 1, 0, 0};
Dot[R[1, -π/2, 0, 0], y]
```

```
{0, 0, 0, 1}
```

As you can check from Figure 1, we get $Y_{1,1}$.

Rotate $Y_{1,0}$ 90 degrees clockwise about the z axis

```
y = {0, 0, 1, 0};
Dot[R[1, -π/2, 0, 0], y]
```

```
{0, 0, 1, 0}
```

As you can check from Figure 1, $Y_{1,1}$ is invariant under rotation about z.

Rotate $Y_{1,1}$ 90 degrees counter – clockwise about the z axis

```
y = {0, 0, 0, 1};
Dot[R[1, π/2, 0, 0], y]
```

```
{0, 1, 0, 0}
```

As you can check from Figure 1, we get $Y_{1,-1}$.

Convert to the axis-angle formalism

...so it's easier to do rotations about $\hat{x}$ and $\hat{y}$.

```
P[u_, θ_] := {
    {Cos[θ] + u[[1]]^2 (1 - Cos[θ]),
     u[[1]] u[[2]] (1 - Cos[θ]) - u[[3]] Sin[θ],
     u[[1]] u[[3]] (1 - Cos[θ]) + u[[2]] Sin[θ]},
    {u[[2]] u[[1]] (1 - Cos[θ]) + u[[3]] Sin[θ],
     Cos[θ] + u[[2]]^2 (1 - Cos[θ]),
     u[[2]] u[[3]] (1 - Cos[θ]) - u[[1]] Sin[θ]},
    {u[[3]] u[[1]] (1 - Cos[θ]) - u[[2]] Sin[θ],
     u[[3]] u[[2]] (1 - Cos[θ]) + u[[1]] Sin[θ],
     Cos[θ] + u[[3]]^2 (1 - Cos[θ])}
   };
```

```
α[u_, θ_] := ArcTan[P[u, θ][[1, 3]], P[u, θ][[2, 3]]];
β[u_, θ_] := ArcTan[P[u, θ][[2, 2]], Sqrt[1 - P[u, θ][[2, 2]]^2]];
γ[u_, θ_] := ArcTan[-P[u, θ][[3, 1]], P[u, θ][[3, 2]]];
```

Rotate $Y_{1,0}$ 90 degrees away from us about the *x* axis

```
y = {0, 0, 1, 0};
u = {1, 0, 0};
θ = -π/2;
Dot[R[1, α[u, θ], β[u, θ], γ[u, θ]], y]
```

```
{0, 1, 0, 0}
```

We get $Y_{1,-1}$, as expected!

Rotate $Y_{1,1}$ 90 degrees about the *y* axis

```
y = {0, 0, 1, 0};
u = {0, 1, 0};
θ = -π/2;
Dot[R[1, α[u, θ], β[u, θ], γ[u, θ]], y]
```

```
{0, 0, 1, 0}
```

We get $Y_{1,0}$, as expected!