**Link to Mathematica .nb file**

# Definitions

## Compute our polynomial basis **bp**

```
bp[n_, x_, y_, z_] := Module[{l, m, μ, ν},
   l = Floor[√n];
   m = n - l^2 - l;
   μ = l - m;
   ν = l + m;
   If[EvenQ[ν], x^(μ/2) y^(ν/2), x^((μ-1)/2) y^((ν-1)/2) z]];
```

## Compute our greens basis **bg**

```
bg[n_, x_, y_, z_] := Module[{l, m, μ, ν},
   l = Floor[√n];
   m = n - l^2 - l;
   μ = l - m;
   ν = l + m;
   Which[
    EvenQ[ν], (μ + 2)/2 x^(μ/2) y^(ν/2),
    ν == 1 && μ == 1, z,
    μ > 1, z ((μ - 3)/2 x^((μ-5)/2) y^((ν-1)/2) - (μ - 3)/2 x^((μ-5)/2) y^((ν+3)/2) - (μ + 3)/2 x^((μ-1)/2) y^((ν-1)/2)),
    OddQ[l], z (-x^(l-3) + x^(l-1) + 4 x^(l-3) y^2),
    True, 3 x^(l-2) y z
   ]];
```

## Compute the greens vectors in the polynomial basis, **p**

```
p[n_, lmax_] := Module[{g},
   g = bg[n, x, y, z];
   Join[{Evaluate[g /. {z → 0, x → 0, y → 0}]},
    Table[
     Coefficient[g, bp[j, x, y, z]] /. {z → 0, x → 0, y → 0}, {j, 1, (lmax + 1)^2 - 1}]]];
```

## The columns of the *inverse* change of basis matrix $\boldsymbol{A_2}^{-1}$ are just **p**

```
A2Inv[lmax_] :=
  Transpose[Flatten[Table[p[l^2 + l + m, lmax], {l, 0, lmax}, {m, -l, l}], 1]];
```

### To get the actual change of basis matrix, just invert it!

```
A2[lmax_] := Inverse[A2Inv[lmax]];
```

# Examples

### Show the basis up to n=15

```
{Table[bg[n, x, y, z], {n, 0, 15}]} // TableForm
```

| | | | | | | | | | | | |
|---|---|---|---|---|---|---|---|---|---|---|---|
| $1$ | $2x$ | $z$ | $y$ | $3x^2$ | $-3xz$ | $2xy$ | $3yz$ | $y^2$ | $4x^3$ | $(1-4x^2-y^2)z$ | $3x^2y$ |

### Show $A_2$ for $l_{max} = 2$

```
A2[2] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{3} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{3} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

# LaTeXify

## LaTeXify

### Make $A_2$ LaTeX-friendly

```
A2TeX[lmax_] := TeXForm[A2[lmax]];
```

## Print $\boldsymbol{A_2}$ for $l_{max} = 2$

```
A2TeX[2]
```

```
\left(
\begin{array}{ccccccccc}
 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & \frac{1}{3} & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{3} & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```