Link to Mathematica .nb file

Case 2 (v odd)

The problem

Our goal is to find a closed form expression for the integral

$$H^v{}_{un}(\phi, b, r) = \int_{\pi-\phi}^{2\pi+\phi} \cos^u \psi \sin^n \psi \, (1 - r^2 - b^2 - 2\, b\, r \sin\psi)^{\frac{3}{2}} \, d\psi$$

for odd v.

Rearranging

First, note that for b > 0 we can write

$$H^v{}_{un}(\phi, b, r) = (2\, br)^{\frac{3}{2}} \int_{\pi-\phi}^{2\pi+\phi} \cos^u \psi \sin^n \psi \, (s - \sin\psi)^{\frac{3}{2}} \, d\psi$$

where we define $s = \frac{1-r^2-b^2}{2\,br} = 2\,k^2 - 1$, with k^2 defined in Appendix A1.

Next, with some algebraic sleight of hand we can express the term in parentheses as

$$(s - \sin\psi)^{\frac{3}{2}} = i(1 - s)^{\frac{3}{2}} \Delta^3$$

where

$$\Delta = \sqrt{1 - \chi^2 \sin^2 x}$$

with

$$\chi^2 = \frac{2}{1-s} = \frac{1}{1-k^2}$$

and

$$x = \frac{\pi}{4} - \frac{\psi}{2}$$

We can check that the real parts of these expressions are equal. Let’s compute the maximum difference for given values of s between (say) -5 and 5:

```
Max[Abs[Table[
    Re[(s - Sin[ψ])^(3/2)] - Re[ⅈ (1 - s)^(3/2) (1 - 2/(1 - s) Sin[π/4 - ψ/2]^2)^(3/2)],
    {ψ, π/2, 2 π + π/2, 0.01},
    {s, -5, 5, 0.0333}]]]
7.10543 × 10^-15
```

There are two caveats:

(1) our expression diverges when b = 0 (s = 1)
(2) the imaginary parts of the two expressions have different signs

We address the first point by noting that when b = 0, the term to the 3/2 power in the *H* integral factors out, so we're back to Case 1 (*v* even), which is easy to solve. The second point ends up not mattering, since the solution to the *H* integral *has* to be real (since it represents a real flux), so the imaginary parts will always cancel.

So let's rewrite the integral as

$$H^{v}{}_{\mathrm{un}}(\phi,\ b,\ r) = i\,(2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\ \frac{3}{2}} \int_{\pi-\phi}^{2\pi+\phi} \cos^{u}\psi \sin^{n}\psi\, \Delta(\psi)^{3}\, d\psi$$

We can do the substitution $\psi \to$x in the *H* integral:

$$\begin{aligned} H^{v}{}_{\mathrm{un}}(\phi,\ b,\ r) &= 2\,i\,(2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\ \frac{3}{2}} \int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} \cos^{u}\left(\tfrac{\pi}{2}-2\,x\right) \sin^{n}\left(\tfrac{\pi}{2}-2\,x\right) \Delta(x)^{3}\, dx \\ &= 2\,i\,(2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\ \frac{3}{2}} \int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} \sin^{u}(2\,x) \cos^{n}(2\,x)\, \Delta(x)^{3}\, dx \\ &= 2\,i\,(2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\ \frac{3}{2}} \int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} (2\cos x \sin x)^{u} \left(\cos^{2} x - \sin^{2} x\right)^{n} \Delta(x)^{3}\, dx \\ &= 2^{u+1}\, i\,(2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\ \frac{3}{2}} \int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} (\cos x \sin x)^{u} \left(\cos^{2} x - \sin^{2} x\right)^{n} \Delta(x)^{3}\, dx \end{aligned}$$

We can use the binomial theorem to expand the $\left(\cos^{2} x - \sin^{2} x\right)^{n}$ term :

$$H^{v}{}_{\mathrm{un}}(\phi, \mathrm{b}, \mathrm{r}) = 2^{u+1} i\ (2\,\mathrm{br})^{\frac{3}{2}}\,(1-s)^{\frac{3}{2}} \sum_{i=0}^{n} (-1)^{n-i} \binom{n}{i} \int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} \cos^{u+2\,i} x\, \sin^{u+2\,n-2\,i} x\, \Delta^{\frac{3}{2}}\, dx$$

And we can do some rearranging to write

$$H^{v}{}_{\mathrm{un}}(\phi, \mathrm{b}, \mathrm{r}) = 2^{u+3}\,(\mathrm{br})^{\frac{3}{2}} \sum_{i=0}^{n} (-1)^{i-n-u} \binom{n}{i} J_{u+2\,i,\,u+2\,n-2\,i}$$

where

$$J_{pq} = i\,(1-s)^{\frac{3}{2}}\,(-1)^q\,\frac{1}{\sqrt{2}}\int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} \cos^p x\,\sin^q x\,\Delta^{\frac{3}{2}}\,dx$$

$$= 2\,i\left(1-k^2\right)^{\frac{3}{2}}\,(-1)^q\int_{-\frac{\phi}{2}-\frac{3\pi}{4}}^{\frac{\phi}{2}-\frac{\pi}{4}} \cos^p x\,\sin^q x\,\Delta^{\frac{3}{2}}\,dx$$

The solution

The reason we rearranged into the form above is that integrals of the form

$$\int \cos^p x \sin^q x\,\Delta^{\frac{3}{2}}\,dx$$

are analytic and solvable via recursion relations (Gradshteyn & Ryzhik 5th edition p.192 #2.581). As we will see, in order for the recurrence to work, we need to have values for J_{00}, J_{02}, J_{20}, J_{22}. As it happens, odd q integrates to zero, while odd p integrates to something nonzero but is never actually needed in the recursion! So let's only bother with J for even p and even q. Let's start by defining some elliptic functions.

```
E1[k2_] := If[k2 < 1,
   (1 - k2) EllipticK[k2],
   (1 - k2)/Sqrt[k2] EllipticK[1/k2]];
```

```
E2[k2_] := If[k2 < 1,
   EllipticE[k2],
   Sqrt[k2] EllipticE[1/k2] + (1 - k2)/Sqrt[k2] EllipticK[1/k2]];
```

where k2 = k^2.

Now, from the equations in Gradshteyn & Ryzhik and with a lot of help from Mathematica, we can write our initial conditions:

```
J00[k2_] := ((8 - 12 k2)/3) E1[k2] + ((-8 + 16 k2)/3) E2[k2];
J02[k2_] := ((8 - 24 k2)/15) E1[k2] + ((-8 + 28 k2 + 12 k2^2)/15) E2[k2];
J20[k2_] := ((32 - 36 k2)/15) E1[k2] + ((-32 + 52 k2 - 12 k2^2)/15) E2[k2];
J22[k2_] := ((32 - 60 k2 + 12 k2^2)/105) E1[k2] + ((-32 + 76 k2 - 36 k2^2 + 24 k2^3)/105) E2[k2];
```

Note that I skipped some steps here. The solutions include sine and cosine terms, but these all cancel in the definite integrals!

We can easily check these expressions for $k^2 > 0$ by comparing to the *real part* of the direct numerical integration of the definition of Jpq:

```
ϕ[k2_] := If[-1 < 2 k2 - 1 < 1, ArcSin[2 k2 - 1], π/2];
JNum[p_, q_, k2_] :=
  Re[NIntegrate[2 ⅈ (1 - k2)^(3/2) (-1)^q Cos[x]^p Sin[x]^q (1 - 1/(1 - k2) Sin[x]^2)^(3/2),
    {x, -ϕ[k2]/2 - 3 π/4, ϕ[k2]/2 - π/4}]];
{{Plot[{JNum[0, 0, With[{foo = k2}, foo]], J00[k2]},
    {k2, 0, 2}, PlotStyle → {Dashing[None], Dashing[Large]}],
   Plot[{JNum[0, 2, With[{foo = k2}, foo]], J02[k2]}, {k2, 0, 2},
    PlotStyle → {Dashing[None], Dashing[Large]}]},
  {Plot[{JNum[2, 0, With[{foo = k2}, foo]], J20[k2]}, {k2, 0, 2},
    PlotStyle → {Dashing[None], Dashing[Large]}],
   Plot[{JNum[2, 2, With[{foo = k2}, foo]], J22[k2]}, {k2, 0, 2},
    PlotStyle → {Dashing[None], Dashing[Large]}]}} // TableForm
```



Finally, we get to the recurrence relation, from Gradshteyn & Ryzhik:

```
J[p_, q_, k2_] := Which[
  OddQ[p], 0,
  OddQ[q], 0,
  (p == 0 && q == 0), J00[k2],
  (p == 0 && q == 2), J02[k2],
  (p == 2 && q == 0), J20[k2],
  (p == 2 && q == 2), J22[k2],
  True, Module[{d1, d2, d3, d4},
   d1 = q + 2 + (p + q - 2) (1 - k2);
   d2 = - (q - 3) (1 - k2);
   d3 = 2 p + q - (p + q - 2) (1 - k2);
   d4 = - (p - 3) + (p - 3) (1 - k2);
   Which[
    q ≥ 4, (d1 J[p, q - 2, k2] + d2 J[p, q - 4, k2]) / (p + q + 3),
    p ≥ 4, (d3 J[p - 2, q, k2] + d4 J[p - 4, q, k2]) / (p + q + 3),
    True, None]
  ]
 ]
```

Let’s again compare this to the numerical solutions:

```
Table[
  With[{r = 0.5}, Plot[{Evaluate[JNum[p, q, k2]], Evaluate[J[p, q, k2]]}, {k2, 0., 2.},
    PlotStyle → {Dashing[None], Dashing[Small]}, Frame → True, Axes → False,
    AspectRatio → 1, FrameTicks → None, PlotLabel → StringForm["p=``, q=``", p, q],
    ImageSize → Tiny]], {p, 0, 6, 2}, {q, 0, 6, 2}
 ] //
 TableForm
```



Woot.