Link to Mathematica .nb file

Definitions

Compute our polynomial basis **bp**

In[1]:=
```
bp[n_, x_, y_] := Module[{l, m, μ, ν},
   l = Floor[Sqrt[n]];
   m = n - l^2 - l;
   μ = l - m;
   ν = l + m;
   If[EvenQ[ν], x^(μ/2) y^(ν/2), x^((μ-1)/2) y^((ν-1)/2) Sqrt[1 - x^2 - y^2]]];
```

Compute our greens basis **bg**

In[2]:=
```
bg[n_, x_, y_] := Module[{l, m, μ, ν},
   l = Floor[Sqrt[n]];
   m = n - l^2 - l;
   μ = l - m;
   ν = l + m;
   Which[
    EvenQ[ν], (μ + 2)/2 x^(μ/2) y^(ν/2),
    ν == 1 && μ == 1, Sqrt[1 - x^2 - y^2],
    μ > 1, Sqrt[1 - x^2 - y^2] ((μ - 3)/2 x^((μ-5)/2) y^((ν-1)/2) - (μ - 3)/2 x^((μ-5)/2) y^((ν+3)/2) - (μ + 3)/2 x^((μ-1)/2) y^((ν-1)/2)),
    OddQ[l], Sqrt[1 - x^2 - y^2] (-x^(l-3) + x^(l-1) + 4 x^(l-3) y^2),
    True, 3 x^(l-2) y Sqrt[1 - x^2 - y^2]
   ]];
```

Compute the greens vectors in the polynomial basis, **p**

In[3]:=
```
p[n_, lmax_] := Module[{g},
   g = bg[n, x, y];
   Join[{Evaluate[g /. {Sqrt[1 - x^2 - y^2] → 0, x → 0, y → 0}]},
    Table[Coefficient[g, bp[j, x, y]] /. {Sqrt[1 - x^2 - y^2] → 0, x → 0, y → 0},
     {j, 1, (lmax + 1)^2 - 1}]]];
```

The columns of the *inverse* change of basis matrix $\boldsymbol{A_2^{-1}}$ are just **p**

In[4]:=
```
A2Inv[lmax_] :=
  Transpose[Flatten[Table[p[l^2 + l + m, lmax], {l, 0, lmax}, {m, -l, l}], 1]];
```

To get the actual change of basis matrix, just invert it!

In[5]:=
```
A2[lmax_] := Inverse[A2Inv[lmax]];
```

Examples

Show the basis up to n=15

In[8]:=
```
{Table[bg[n, x, y] /. Sqrt[1 - x^2 - y^2] → z, {n, 0, 15}]} // TableForm
```

Out[8]//TableForm=

$1 \quad 2x \quad z \quad y \quad 3x^2 \quad -3xz \quad 2xy \quad 3yz \quad y^2 \quad 4x^3 \quad (1-4x^2-y^2)z \quad 3x^2y$

Show A_2 for $l_{max} = 2$

In[9]:=
```
A2[2] // MatrixForm
```

Out[9]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \frac{1}{3} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{3} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

In[10]:=

LaTeXify

Out[10]= LaTeXify

Make A_2 LaTeX-friendly

In[11]:=
```
A2TeX[lmax_] := TeXForm[A2[lmax]];
```

Print A_2 for $l_{max} = 2$

In[12]:=
```
A2TeX[2]
```

Out[12]//TeXForm=
```
\left(
\begin{array}{ccccccccc}
 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & \frac{1}{3} & 0 & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & -\frac{1}{3} & 0 & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{2} & 0 & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{3} & 0 \\
 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```